
Bar Charts

Okay, I am the first to admit, there are some pretty sophisticated PC (and Mac) packages out there for displaying data as bar charts, pie charts, three dimensional charts, etc. There is no way Natural running on a mainframe can match the array of options on a PC. 

That said, there are many ways to use Natural to display data in a manner that is far more attractive than green screen spreadsheets. In particular, Natural has facilities that permit the creation of really attractive bar charts.

I have used the DY facility with great effectiveness to create bar charts. As mentioned earlier, in a forthcoming issue, I will discuss DY in some depth. Here, we will only discuss the rather simple use of DY required for purposes of our example.

Real applications

First, here is a typical “green screen” DISPLAY of some data.

DEFINE DATA LOCAL
1 #SALES (N4/1:8,1:6)
  INIT (1,V) <122,186,165,301,225,272>
       (2,V) <175,183,225,192,228,230>
       (3,V) <142,151,165,180,192,268>
       (4,V) <133,142,224,204,183,215>
       (5,V) <221,233,173,215,237,208>
       (6,V) <135,155,142,168,199,231>
       (7,V) <206,214,227,242,185,244>
       (8,V) <188,195,160,182,159,202>
1 #LOOP1 (P3)
1 #PRODUCT (A11)
END-DEFINE
*
FOR #LOOP1 = 1 TO 8
COMPRESS 'PRODUCT' #LOOP1 INTO #PRODUCT
DISPLAY 5T 'PRODUCT' #PRODUCT
        '2001' #SALES (#LOOP1,1) '2002' #SALES (#LOOP1,2)
        '2003' #SALES (#LOOP1,3) '2004' #SALES (#LOOP1,4)
        '2005' #SALES (#LOOP1,5) '2006' #SALES (#LOOP1,6)

END-FOR
END

We will assume, for purposes of the discussion, that we have read in sales data for eight products (imaginatively named PRODUCT 1 through PRODUCT 8) for each of six years, 2001 – 2006. Here is the output for the program above.

    PAGE #   1                    DATE:    MAR 02, 2009
    PROGRAM: BAR03X               LIBRARY: INSIDE

      PRODUCT   2001  2002  2003  2004  2005  2006
    ----------- ----- ----- ----- ----- ----- -----

    PRODUCT 1     122   186   165   301   225   272
    PRODUCT 2     175   183   225   192   228   230
    PRODUCT 3     142   151   165   180   192   268
    PRODUCT 4     133   142   224   204   183   215
    PRODUCT 5     221   233   173   215   237   208
    PRODUCT 6     135   155   142   168   199   231
    PRODUCT 7     206   214   227   242   185   244
    PRODUCT 8     188   195   160   182   159   202

Okay, all the data is there. However, if I wanted to “see” patterns of increase or decline, the spreadsheet display of the data is hardly helpful. 

Here is a program that takes advantage of Natural’s color and reverse video capabilities to present the data in a format that could potentially be far more useful.

DEFINE DATA LOCAL
1 #SALES (N4/1:8,1:6)
  INIT (1,V) <122,186,165,301,225,272>
(2,V) <175,183,225,192,228,230>
(3,V) <142,151,165,180,192,268>
(4,V) <133,142,224,204,183,215>
(5,V) <221,233,173,215,237,208>
(6,V) <135,155,142,168,199,231>
(7,V) <206,214,227,242,185,244>
(8,V) <188,195,160,182,159,202>
1 REDEFINE #SALES
  2 #SALES-A (A4/1:8,1:6)
1 #LOOP1 (P3)
1 #LOOP2 (P3)
1 #MAX   (P3)
1 #NUM (P3)
1 #SPOT (P3)
1 #BAR-LINE (A75)
1 REDEFINE #BAR-LINE
  2 #BAR (A1/1:75)
1 #SCALE (P5.5)
1 #YEAR (N4)
1 #YEAR-LINE (A75)
1 #HOLD (P4)
END-DEFINE
INCLUDE AASETC
INCLUDE AATITLER
* FIRST, COMPUTE MAX OF ALL OUR DATA POINTS
*
FOR #LOOP1 = 1 TO 8
  FOR #LOOP2 = 1 TO 6
      IF #SALES (#LOOP1,#LOOP2) GT #MAX
      MOVE #SALES (#LOOP1,#LOOP2) TO #MAX
      END-IF
  END-FOR
END-FOR
*
* NOW, COMPUTE SCALE FACTOR
*
COMPUTE #SCALE = #MAX / 12
*
FOR #LOOP1 = 1 TO 8
    IF #LOOP1 = 5
      NEWPAGE
      END-IF
  RESET #SPOT
  FOR #LOOP2 = 1 TO 6
    ADD 1 TO #SPOT
    COMPUTE #YEAR = 2000 + #LOOP2
    MOVE '@' TO #BAR (#SPOT)
    ADD 1 TO #SPOT
    MOVE #SALES (#LOOP1,#LOOP2) TO 
         SUBSTRING (#BAR-LINE,#SPOT,4)
/*
    MOVE #YEAR TO SUBSTRING (#YEAR-LINE,#SPOT,4)
    COMPUTE #NUM = #SALES (#LOOP1,#LOOP2) / #SCALE
    ADD #NUM  TO #SPOT
    MOVE '#' TO #BAR (#SPOT)
  END-FOR
  WRITE 1X 'PRODUCT NUMBER:' #LOOP1 (NL=1) / 
        #BAR-LINE (DY=@VYE#) / #YEAR-LINE
  RESET #BAR-LINE #YEAR-LINE
  SKIP 2
END-FOR
*
END

The DY for #BAR-LINE is actually quite similar to our earlier use of DY. When Natural is WRITE’ing #BAR-LINE, it looks for ampersands. If it finds any, Natural switches output to reverse video yellow. If Natural finds any # marks, it will revert to normal display. 

Yes,  in a real “system” I would have to check to see if any of the reverse video bars representing sales for a given year were less than four characters wide. This is not difficult to do, and could also be addressed by truncating the first two characters of the year.

Here is the first page of our output:

Unfortunately, Inside Natural is not in color. You will have to run the program to appreciate how user friendly the output is.

Speaking of color, I decided that having all the blocks the same color was “boring”. So I thought about alternating colors. I soon realized that this was messier than I thought, especially when I saw the results of one piece of code with two colors, and contemplated what would be required to switch to three colors. So, I thought about the code for switching colors and came up with a technique that works quite well, regardless of how many colors one wants to utilize.

DEFINE DATA LOCAL
1 #SALES (N4/1:8,1:6)
  INIT (1,V) <122,186,165,301,225,272>
       (2,V) <175,183,225,192,228,230>
       (3,V) <142,151,165,180,192,268>
       (4,V) <133,142,224,204,183,215>
       (5,V) <221,233,173,215,237,208>
       (6,V) <135,155,142,168,199,231>
       (7,V) <206,214,227,242,185,244>
       (8,V) <188,195,160,182,159,202>
1 REDEFINE #SALES
  2 #SALES-A (A4/1:8,1:6)
1 #LOOP1 (P3)
1 #LOOP2 (P3)
1 #MAX   (P3)
1 #NUM (P3)
1 #SPOT (P3)
1 #BAR-LINE (A75)
1 REDEFINE #BAR-LINE
  2 #BAR (A1/1:75)
1 #SCALE (P5.5)
1 #YEAR (N4)
1 #YEAR-LINE (A75)
1 #HOLD (P4)
1 #START (A1/1:6) INIT <'@','*','@','*','@','*'>
1 #END (A1/1:6)INIT <'#','&','#','&','#','&'>
END-DEFINE
*
* FIRST, COMPUTE MAX OF ALL OUR DATA POINTS
*
FOR #LOOP1 = 1 TO 8
    FOR #LOOP2 = 1 TO 6
    IF #SALES (#LOOP1,#LOOP2) GT #MAX
       MOVE #SALES (#LOOP1,#LOOP2) TO #MAX
       END-IF
    END-FOR
END-FOR
*
* NOW, COMPUTE SCALE FACTOR
*
COMPUTE #SCALE = #MAX / 12
*
FOR #LOOP1 = 1 TO 8
IF #LOOP1 = 5
   NEWPAGE
   END-IF
RESET #SPOT
    FOR #LOOP2 = 1 TO 6
    ADD 1 TO #SPOT
    COMPUTE #YEAR = 2000 + #LOOP2
    MOVE #START (#LOOP2) TO #BAR (#SPOT)
    ADD 1 TO #SPOT
    MOVE #SALES (#LOOP1,#LOOP2) TO 
         SUBSTRING (#BAR-LINE,#SPOT,4)
    MOVE #YEAR TO SUBSTRING (#YEAR-LINE,#SPOT,4)
    COMPUTE #NUM = #SALES (#LOOP1,#LOOP2) / #SCALE
    ADD #NUM  TO #SPOT
    MOVE #END (#LOOP2) TO #BAR (#SPOT)
    END-FOR
WRITE 1X 'PRODUCT NUMBER:' #LOOP1 (NL=1) / #BAR-LINE  
         (DY=@VYE#*VRE&) / #YEAR-LINE
RESET #BAR-LINE #YEAR-LINE
SKIP 2
END-FOR
*
END

The “key” to the beauty of this code are the two arrays, #START and #END. By simply changing the values in this array, and the DY specification in the WRITE statement, you can alter the number of colors. For example:

1 #START (A1/1:6) INIT <'@','*','%','@','*','%'>
1 #END (A1/1:6)INIT <'#','&','$','#','&','$'>

WRITE 1X 'PRODUCT NUMBER:' #LOOP1 (NL=1) / 
         #BAR-LINE (DY=@VYE#*VRE&%VTU$) / #YEAR-LINE


Just make these changes to the program above, and, presto, three colors instead of two.

